
STM32H7

Vhodno / izhodne naprave

USART Serijska komunikacija

z uporabo DMA krmilnika

http://www.fri.uni-lj.si/si

STM32H750XB

http://www.fri.uni-lj.si/si

Priključitev :

• Mikro USB priklop na daljši stranici (nad LCD, srednji !!!)

Poseben začetni projekt (github) in info za STM32H7 (e-učilnica):

• dodajanje vsebine (Main.s):

Delo na STM32H7 razvojnem sistemu

http://www.fri.uni-lj.si/si

STM32H750B-DK - Schematic

http://www.fri.uni-lj.si/si

U(S)ART in DMA

naprave

http://www.fri.uni-lj.si/si

DMA- Splošno

Primer 1: CPE opravlja prenose V/I <-> pomnilnik

• Čaka na zastavico

• ldr Rx, … in str Rx, …

Primer 2: DMA opravlja prenose V/I <-> pomnilnik :

• CPE nastavi DMA krmilnik za prenos :

• Vrsta prenosa, naprave

• Naslovi vira in ponora

• Velikost in število podatkov

• Sproži prenos

• DMA krmilnik (neodvisno od CPE):

• Čaka na zastavico

• Prebere in shrani podatek

• odšteva preostale podatke in zaključi prenos

• CPE izvaja svoj program

http://www.fri.uni-lj.si/si

DMA- Splošno

Case1: In this example, filling the first and the third buffer took the exactly the same time, while copying the first buffer

to the second one took slightly less time:

While the DMA cannot be used to compute Fibonacci numbers, or initialize arrays with non-constant values, it can be 

used for copying data between 2 memory locations. 

Now the DMA operation ran in parallel with the CalculateFibonacci() function, reducing the overall program time by

21%:

Case2: CRC calculation case:

The results are in:

80 DMA CRCs per second. 

63 manual CRCs per second

On my processor, DMA gives a 27% advantage over iterative memory assignment. I think it is because everything

is done with a hardware mover that doesn’t have to increment, involve registers, gotos, branch less than, and so on.

http://www.fri.uni-lj.si/si
https://visualgdb.com/w/wp-content/uploads/2019/06/timing1.png
https://visualgdb.com/w/wp-content/uploads/2019/06/timing2.png

FRI-SMS: DMA Krmilnik (PDC – Peripheral DMA Controller)

http://www.fri.uni-lj.si/si

• Za prenos podatkov, ki ne troši CPE časa

• 22 DMA kanalov

• V/I naprave z DMA kanali

• DBGU, SPI, USART, SSC, MCI, EMAC, ISI, ADC

• Hkratno dvosmerne V/I naprave (full duplex) imajo dva DMA kanala

• Enosmerne in izmenično dvosmerne (half duplex) V/I naprave imajo po 
en DMA kanal

• Preprosto programiranje. Potrebno je vpisati le:

• začetni naslov in

• dolžino bloka za prenos

• DMA krmilnik je dostopen preko naslovnega prostora vsake naprave 
posebej od odmika 0x100 dalje

FRI-SMS: DMA Krmilnik

http://www.fri.uni-lj.si/si

Primer 1 (brez DMA): CPE opravlja prenose V/I <-> pomnilnik

• Čaka na zastavico

• ldr Rx, … in str Rx, …

Primer 2 (z DMA): DMA opravlja prenose V/I <-> pomnilnik :

• CPE nastavi DMA krmilnik za prenos :

• DMA krmilnik (neodvisno od CPE):

• CPE izvaja svoj program

DMA- STM32H7 (stikalna matrika)

http://www.fri.uni-lj.si/si

DMA - STM32H7

http://www.fri.uni-lj.si/si

DMA - STM32H7 + USART3

http://www.fri.uni-lj.si/si

Viri USART3 : User & Programming manuals, vezalna shema

Vezalna shema

https://st-onlinetraining.s3.amazonaws.com/STM32H7-Peripheral-USART-interface_%28USART%29/index.html

Ponovitev OR-LAB 10: USART

http://www.fri.uni-lj.si/si
https://st-onlinetraining.s3.amazonaws.com/STM32H7-Peripheral-USART-interface_%28USART%29/index.html

USART – krmiljenje 
1. INIT_USART3 - Inicializacija USART3 naprave  (prejšnja LAB vaja)

Potrebni koraki za krmiljenje USART naprave:

1. Vklop USART3 naprave

• RCC_APB1LENR : b18=1 (USART3 Enable Clock)

2. Nastavitev GPIOB priključkov 10,11 na AF7(Alt. Function7)

• RCC_AHB4ENR (Peripheral Clock Register): b1=1 .. Port B Enable

• GPIOB_MODER (Mode Register): 0b10,  AF on pins PB10,PB11 

• GPIOB_AFRH (AF Register): 0x07700 AF7 on pins PB10,PB11 

3. Nastavitev hitrosti delovanja (BaudRate)

• USART3_BRR (BaudRate Register): 64M/115200 ≈ 556

4. Sprožitev delovanja

• USART3_CR1 (Control Register 1): 0b1101 TX, RX, USART Enable bits

5. Delovanje

Oddaja znaka: 

• USART3_ISR:  ko TXE=1, vpis znaka v USART3_TDR

Sprejem znaka: 

• USART3_ISR:  ko RXNE=1, preberi znak iz USART3_RDR

DMA omogoča delo z nizi znakov !

http://www.fri.uni-lj.si/si

Vir DMA: Reference manual

Razlike v primerjavi s Cortex M4 (STM32F4):

• DMAMUX omogoča poljubne preslikave 

naprav na 16 DMA kanalov

• DMA napravi (DMA1,DMA2) delujeta 

samo v SRAM pomnilnik(e)

• Pozor pri zagonu kode iz RAM-a !!

http://www.fri.uni-lj.si/si

USART + DMA – krmiljenje 
Potrebni koraki za krmiljenje USART naprave s pomočjo DMA naprave :

1. INIT_USART3 - Inicializacija USART3 naprave 

• Enako kot za samostojno delovanje naprave USART3 (OR Vaja 10)
• za sprejem ali oddajo posameznega znaka

2. INIT_DMA - Inicializacija DMA in dodatne nastavitve USART3

• RCC: vklop DMA1 naprave

• Izklop DMA kanalov (RX, TX) in njune nastavitve
• Splošne nastavitve za prenose : 

– smer prenosa (pomn.<-> naprava), povečevanje naslovov v pomnilniku

• DMAMUX1: 
• Možnost poljubne preslikave med V/I napravami in kanali DMA krmilnika (angl. stream)

• Nastavitve DMA kanalov 1,0 in povezava z napravo USART3 (št. 45 = RX, št. 46 = TX)

• Izklop USART3 naprave in vklop DMA krmiljenja ter USART3:
• Dodatna nastavitev USART3- vklop DMA krmiljenja prenosov na USART3 napravi

3. SND_DMA, RCV_DMA: Nastavitve DMA za vsak prenos 

• za sprejem ali oddajo niza znakov
• Naslov podatkovnega registra V/I naprave, pomnilnika in št. znakov za prenos

• Brisanje zastavic za prejšnje dogodke

• Vklop ustreznega kanala DMA (TX ali RX) 

http://www.fri.uni-lj.si/si

2. INIT_DMA - Inicializacija DMA in USART3
RCC: vklop DMA1 naprave

// Enable DMA1 Peripheral Clock (bit 0 in AHB1ENR 
register)
ldr r6, =RCC_BASE          // Load peripheral clock reg 
base address to r6
ldr r5, [r6,#RCC_AHB1ENR]   // Read its content to r5
orr r5, #1              // Set bit 0 to enable DMA1 clock
str r5, [r6,#RCC_AHB1ENR]   // Store result in peripheral
clock register

http://www.fri.uni-lj.si/si

Izklop DMA kanalov (RX, TX) in njune nastavitve

.equ     DMA_USART3_TX_STREAM, 0    //Channel 0 on DMA1

.equ     DMA_USART3_RX_STREAM, 1    //Channel 1 on DMA1

2. INIT_DMA - Inicializacija DMA in USART3

ldr r6, =DMA1_BASE    // Load DMA1 BASE address to r6

// ------- Primer: DMA1 TX Settings
// Disable DMA TX Channel
ldr r5, [r6,#DMA_SxCR_TX] // Read its content to r5
bic r5, #1
str r5, [r6,#DMA_SxCR_TX] // Store result

wt0_EN0:
ldr r5, [r6,#DMA_SxCR_TX] // wait until bit is read as 0
tst r5, #1
bne wt0_EN0

http://www.fri.uni-lj.si/si

Default nastavitve, ki 

ustrezajo

Nastavitve za 

spremembo

b10=1
• povečevanje 

naslova v pomn.

b7,6=
• 01 TX

• 00 RX

2. INIT_DMA - Inicializacija DMA in USART3

// ------- Primer: DMA1 TX Settings
// Set MINC (increment memory pointer) and Direction 
(Memory->Peripheral)
orr r5,r5,#(0b10001 << 6)    // b10=1 and bit7,6 = 01
str r5, [r6,#DMA_SxCR_TX]     // Store result

Izklop DMA kanalov (RX, TX) in njune nastavitve

http://www.fri.uni-lj.si/si

Default nastavitve, ki 

ustrezajo (podrobneje)

2. INIT_DMA - Inicializacija DMA in USART3
Izklop DMA kanalov (RX, TX) in njune nastavitve

http://www.fri.uni-lj.si/si

Nastavitev direktnega načina (že nastavljeno na 0)

2. INIT_DMA - Inicializacija DMA in USART3

// ------- Primer: DMA1 TX Settings
// Enable direct mode
ldr r5, [r6,#DMA_SxFCR_TX]   // Read its content to r5
bic r5, #0b100
str r5, [r6,#DMA_SxFCR_TX]     // Store result

Default nastavitve, ki 

ustrezajo

Izklop DMA kanalov (RX, TX) in njune nastavitve

http://www.fri.uni-lj.si/si

Nastavitev preslikave DMAMUX1 (naprava <-> stream)

2. INIT_DMA - Inicializacija DMA in USART3
DMAMUX1: 

http://www.fri.uni-lj.si/si

Nastavitev preslikave DMAMUX1 (naprava <-> stream)

2. INIT_DMA - Inicializacija DMA in USART3

// ------------- DMAMUX1 Settings
// Set channels to devices translations (multiplexing)

ldr r6, =DMAMUX1_BASE         // Load reg base address to r6

mov r5,#46 // USART3_TX DMA Device Nr. is 46
str r5, [r6, DMAMUX1_C0CR]// DMAREQ for Channel 0 to USART3_TX

mov r5,#45 // USART3_Rx DMA Device Nr. is 45
str r5, [r6, DMAMUX1_C1CR]// DMAREQ for Channel 1 to USART3_RX

DMAMUX1: 

http://www.fri.uni-lj.si/si

Izklop, vklop delovanja USART 3 (za konfiguracijo CR3)

2. INIT_DMA - Inicializacija DMA in USART3
Izklop USART3 naprave in vklop DMA krmiljenja ter USART3:

http://www.fri.uni-lj.si/si


OR – Organizacija računalnikov 25

// ------------- USART3 Settings
ldr r6, =USART3_BASE       // Load USART3 BASE address to r1

// Disable USART3
ldr r5, [r6,#USART_CR1]   // Read its content to r5
bic r5, #1
str r5, [r6,#USART_CR1]   // Store result

// Enable DMA Transmit and Receive for USART3
ldr r5, [r6, #USART_CR3]
orr r5, #(0b11<<6)          // Set bits 7 and 6 to enable DMAT and DMAR bits
str r5, [r6,#USART_CR3]    // Store result

// Enable USART3
ldr r6, =USART3_BASE       // Load USART3 BASE address to r6
ldr r5, [r6,#USART_CR1]   // Read its content to r5
orr r5, r5, #1
str r5, [r6,#USART_CR1]   // Store result

2. INIT_DMA - Inicializacija DMA in USART3

Vklop DMA prenosov na USART3

Izklop USART3 naprave in vklop DMA krmiljenja ter USART3:

http://www.fri.uni-lj.si/si

3. RCV_DMA ,SND_DMA: Nastavitve DMA 

Nastavitev naslova USART3 naprave (Data Register)

Naslov USART3 Data 

registra:

USART3_BASE+USART_RDR

ldr r6, =DMA1_BASE          // Load reg base address to r6

WAIT_EN: // Wait EN bit to become zero
ldr r5, [r6, #DMA_SxCR_RX]
tst r5, #1
bne WAIT_EN

//  Receive (RX) DMA Init
ldr r5, =USART3_BASE+USART_RDR// RX peripheral address to r5
str r5, [r6,#DMA_SxPAR_RX]   // Store peripheral DMA pointer
str r0, [r6,#DMA_SxM0AR_RX]   // Store address pointer in r0
str r1, [r6,#DMA_SxSNDTR_RX]   // Store number of units in r1

http://www.fri.uni-lj.si/si

Nastavitev naslova v pomnilniku (za prenos)

3. RCV_DMA ,SND_DMA: Nastavitve DMA 

Naslov v 

pomnilniku

(začetek 

niza v r0)

ldr r6, =DMA1_BASE          // Load reg base address to r6

//  Receive (RX) DMA Init
ldr r5, =USART3_BASE+USART_RDR// RX peripheral address to r5
str r5, [r6,#DMA_SxPAR_RX]   // Store peripheral DMA pointer
str r0, [r6,#DMA_SxM0AR_RX]   // Store address pointer in r0
str r1, [r6,#DMA_SxSNDTR_RX]   // Store number of units in r1

http://www.fri.uni-lj.si/si

Nastavitev števila podatkov (za prenos)

3. RCV_DMA ,SND_DMA: Nastavitve DMA 

Število 

znakov 

v r1

ldr r6, =DMA1_BASE          // Load reg base address to r6

//  Receive (RX) DMA Init
ldr r5, =USART3_BASE+USART_RDR// RX peripheral address to r5
str r5, [r6,#DMA_SxPAR_RX]   // Store peripheral DMA pointer
str r0, [r6,#DMA_SxM0AR_RX]   // Store address pointer in r0
str r1, [r6,#DMA_SxSNDTR_RX]   // Store number of units in r1

http://www.fri.uni-lj.si/si

Brisanje prejšnjih DMA dogodkov

ldr r6, =DMA1_BASE          // Load reg base address to r6

// Clear flags in Status register
mov r5,#(0b111101<<6)           // clear flags for Ch1 in ISR
str r5, [r6,# DMA_LIFCR]       // Store 

3. RCV_DMA ,SND_DMA: Nastavitve DMA 

http://www.fri.uni-lj.si/si

Vklop delovanja (streama) DMA krmilnika (EN=0)

3. RCV_DMA ,SND_DMA: Nastavitve DMA 

//  Enable DMA Channel
ldr r5, [r6, #DMA_SxCR_RX]
orr r5, r5, #1
str r5, [r6, #DMA_SxCR_RX]// Enable channel

http://www.fri.uni-lj.si/si


OR – Organizacija računalnikov 31

Čakanje na zaključek prenosov

// --------- RCV_DMA
// Wait for the end of reception (TCIF)
   ldr r6, =DMA1_BASE          // Load reg base address to r6

WAIT_RC:
ldr r5, [r6, #DMA_LISR]
tst r5, #(1 << 11)        // TCIF1 flag
beq WAIT_RC

// --------- SND_DMA
// Wait for the end of transmission

ldr r6, =USART3_BASE

WAIT_TC:
ldr r5, [r6, #USART_ISR]

    tst r5, #(1 << 6)              // Test TC bit
beq WAIT_TC 

Sprejem (RCV_DMA):

Oddaja (SND_DMA):

3. RCV_DMA ,SND_DMA: Nastavitve DMA 

http://www.fri.uni-lj.si/si

DMA – stanje , nastavitve

Status reg. 

Števec znakov

Clear

flags reg.

Ctrl reg. #0 

Naslov DReg

Naslov Pomn.

Ctrl reg. #1

Ch0

Ch1
…

DMA

http://www.fri.uni-lj.si/si

DMA – projekt
STM32H750XBHX_RAM.ld:

• Dodamo sekcijo .sram

/* Define output sections */
SECTIONS
{

…
/* used by the startup to initialize data */
_sisram = LOADADDR(.sram); /* this will be start address of init values in RAM_EXEC */

/* Initialized data sections goes into SRAM, have to copy content manually */
.sram :
{
. = ALIGN(4);
_ssram = .; /* create a global symbol at sram start */
KEEP(*(.sram)) /* .data sections */
*(.sram*) /* .data* sections */

. = ALIGN(4);
_esram = .; /* define a global symbol at sram end */

} >RAM_D2 AT> RAM_EXEC
…

}

http://www.fri.uni-lj.si/si

DMA – program
Naslovi registrov:
//-----------------------------
// DMA related definitions
//-----------------------------

// AHB2ENR register offset is 0xDC (for enabling SRAMs)
.equ RCC_AHB2ENR, 0xDC // RCC AHB2ENR periph. clk reg.

// AHB1ENR register offset is 0xE8
.equ RCC_AHB1ENR, 0xD8 // RCC AHB1ENR periph. clk reg.

// DMAMUX1 base address is 0x40020800
.equ DMAMUX1_BASE, 0x40020800 // DMAMUX1 base address

.equ DMAMUX1_C0CR, 0x00 // CR for Channel 0

.equ DMAMUX1_C1CR, 0x04 // CR for Channel 1 

// DMA1 base address is 0x40020000
.equ DMA1_BASE, 0x40020000 // DMA1 base address
// DMA2 base address is 0x40020400
.equ DMA2_BASE, 0x40020400 // DMA2 base address

.equ DMA_USART3_TX_STREAM, 0 //Channel 0 on DMA1

.equ DMA_USART3_RX_STREAM, 1 //Channel 1 on DMA1

// DMA Registers definitions
.equ DMA_LISR, 0x00
.equ DMA_HISR, 0x04
.equ DMA_LIFCR, 0x08
.equ DMA_HIFCR, 0x0C

.equ DMA_SxCR_TX,0x10 + 0x18 * DMA_USART3_TX_STREAM

.equ DMA_SxFCR_TX,0x24 + 0x18 * DMA_USART3_TX_STREAM

.equ DMA_SxSNDTR_TX,0x14 + 0x18 * DMA_USART3_TX_STREAM

.equ DMA_SxPAR_TX,0x18 + 0x18 * DMA_USART3_TX_STREAM

.equ DMA_SxM0AR_TX,0x1C + 0x18 * DMA_USART3_TX_STREAM

.equ DMA_SxCR_RX,0x10 + 0x18 * DMA_USART3_RX_STREAM

.equ DMA_SxFCR_RX,0x24 + 0x18 * DMA_USART3_RX_STREAM

.equ DMA_SxSNDTR_RX,0x14 + 0x18 * DMA_USART3_RX_STREAM

.equ DMA_SxPAR_RX,0x18 + 0x18 * DMA_USART3_RX_STREAM

.equ DMA_SxM0AR_RX,0x1C + 0x18 * DMA_USART3_RX_STREAM

http://www.fri.uni-lj.si/si

DMA – program (zagon z RAM povez. skripto)
Main.s - spremembe:

• Dodamo spremenljivke v sekcijo .sram

// Start of sram section
.section .sram,"a",%progbits
.align

NIZ1: .space 12
NIZ2: .asciz "Testni niz!" // 12 bytes

.equ NIZ_LEN, 12

• Za delovanje moramo vklopiti tudi SRAM pomnilnike :

/* Enable SRAMs and copy initial values only with RAM Linker script */
// Enable SRAMs internal memories (for DMA1)
bl SRAM_ENABLE

• Startup koda ne vsebuje obravnave te sekcije, zato moramo za kopiranje začetnih 

vrednosti poskrbeti sami:

// Copy initial values for .sram section (from .text to SRAM)
ldr r0,=_sisram
ldr r1,=_ssram
ldr r2,=_esram
bl MEM_COPY

MEM_COPY:
push {r3, lr}

copy:
ldr r3,[r0],#4
str r3,[r1],#4

cmp r1,r2
blo copy

pop {r3, pc}

http://www.fri.uni-lj.si/si

DMA – program (zagon z RAM povez. skripto)
Main.s:
// Comment following line when FLASH Linker Script is used
#define RAM_LinkScript
…
#ifdef RAM_LinkScript
// Start of sram section

.section .sram,"a",%progbits
#endif

.align
NIZ1: .space 12

.align
NIZ2: .asciz "Testni niz!" // 12 bytes

.equ NIZ_LEN, 12

main:

#ifdef RAM_LinkScript

// Relocating Vector table to RAM (only for RAM Linker Script)
bl RELLOC_VECTBL

/* Enable SRAMs and copy initial values only with RAM Linker script */
// Enable SRAMs internal memories (for DMA1)

bl SRAM_ENABLE

// Copy initial values for .sram section (from .text to SRAM)
ldr r0,=_sisram
ldr r1,=_ssram
ldr r2,=_esram
bl MEM_COPY

#endif

MEM_COPY:
push {r3, lr}

copy:
ldr r3,[r0],#4
str r3,[r1],#4

cmp r1,r2
blo copy

pop {r3, pc}

SRAM_ENABLE:
push {r5, r6, lr}

// Enable internal SRAMs (bits 31,30,29 in AHBeENR register)
ldr r6, =RCC_BASE // Load periph. clk reg base address to r6
ldr r5, [r6,#RCC_AHB2ENR] // Read its content to r5
orr r5, #(0b111 << 29) // Set bits 31,30,29 to 1 to enable
SRAMs clock
str r5, [r6,#RCC_AHB2ENR] // Store result in periph. clk
register

pop {r5, r6, pc}

RELLOC_VECTBL:

push {r0, r1, lr}

ldr r1, =VTOR // Set Vector table addr. to 
0x24000000
ldr r0, =0x24000000
str r0, [r1]

pop {r0, r1, pc}

http://www.fri.uni-lj.si/si

Vklop SRAM pomnilnikov (DMA1,2 delujeta le v tej domeni)

SRAM_ENABLE:
push {r5, r6, lr}

// Enable internal SRAMs (bits 31,30,29 in AHBeENR register)
ldr r6, =RCC_BASE // Load periph. clk reg base address to r6
ldr r5, [r6,#RCC_AHB2ENR] // Read its content to r5
orr r5, #(0b111 << 29) // Set bits 31,30,29 to 1 to enable SRAMs
clock
str r5, [r6,#RCC_AHB2ENR] // Store result in periph. clk register

pop {r5, r6, pc}

http://www.fri.uni-lj.si/si

DMA – program (glavna zanka)
Main.s:

bl INIT_DMA // Priprava DMA naprave za prenose preko USART3

// Main loop for USART3+DMA Echo test for blocks of 12 characters with change to upper case
loop:

ldr r0,=NIZ2
mov r1,#NIZ_LEN
bl SND_DMA

bl LED_ON // Vklop LED diode

mov r0,#500
// bl DELAY // Zakasnitev SW Delay: r0 x 1msec
bl DELAYTC // Zakasnitev SysTick : r0 x 1msec

ldr r0,=NIZ1
mov r1,#NIZ_LEN
bl RCV_DMA

ldr r0,=NIZ1
ldr r1,=NIZ2
ldr r2, =NIZ_LEN
bl CHANGE

bl LED_OFF // Izlop LED diode

mov r0,#500
// bl DELAY // Zakasnitev SW Delay: r0 x 1msec
bl DELAYTC // Zakasnitev SysTick : r0 x 1msec

b loop // skok na vrstico loop:

CHANGE:
push {r3-r4,lr}

ch_zanka:
ldrb r4, [r0], #1
bic r3, r4, #0b100000 // zbrisi b5

cmp r3, #'A‘
blo pisi

cmp r3, #'Z‘
bhi pisi

eor r4, r4, #0b100000 // spremeni crko

pisi:
strb r4, [r1], #1 /* shranimo v niz2*/

subs r2, r2, #1
bne ch_zanka

pop {r3-r4,pc}

DMA_INIT (1/3 – DMA1, DMA TX): 
INIT_DMA:

push {r5, r6, lr}

//-----------------------------------------
// ------------- DMA1 Settings

// Enable DMA1 Peripheral Clock (bit 0 in AHB1ENR register)
ldr r6, =RCC_BASE // Load peripheral clock reg base address to r6
ldr r5, [r6,#RCC_AHB1ENR] // Read its content to r5
orr r5, #1 // Set bit 0 to enable DMA1 clock
str r5, [r6,#RCC_AHB1ENR] // Store result in peripheral clock register

ldr r6, =DMA1_BASE // Load DMA1 BASE address to r6

// ------- DMA1 TX Settings
// Disable DMA TX Channel
ldr r5, [r6,#DMA_SxCR_TX] // Read its content to r5
bic r5, #1
str r5, [r6,#DMA_SxCR_TX] // Store result

wt0_EN0:
ldr r5, [r6,#DMA_SxCR_TX] // wait until bit is read as 0
tst r5, #1
bne wt0_EN0

// Set MINC (increment memory pointer) and Direction (Memory->Peripheral)
orr r5,r5,#(0b10001 << 6) // b10=1 and bit7,6 = 01
str r5, [r6,#DMA_SxCR_TX] // Store result

// Enable direct mode
ldr r5, [r6,#DMA_SxFCR_TX] // Read its content to r5
bic r5, #0b100
str r5, [r6,#DMA_SxFCR_TX] // Store result

DMA_INIT (2/3 – DMA TX, DMAMUX): 

// ------- DMA1 RX Settings
// Disable DMA RX Channel
ldr r5, [r6,#DMA_SxCR_RX] // Read its content to r5
bic r5, #1
str r5, [r6,#DMA_SxCR_RX] // Store result

wt1_EN0:
ldr r5, [r6,#DMA_SxCR_RX] // wait until bit is read as 0
tst r5, #1
bne wt1_EN0

// Set MINC (increment memory pointer) and Direction (Memory<-Peripheral)
orr r5,r5,#(0b10000 << 6) // b10=1 and bit7,6 = 00
str r5, [r6,#DMA_SxCR_RX] // Store result

// Enable direct mode
ldr r5, [r6,#DMA_SxFCR_RX] // Read its content to r5
bic r5, #0b100
str r5, [r6,#DMA_SxFCR_RX] // Store result

//----------------------------------------------
// ------------- DMAMUX1 Settings
// Set channels to devices translations (multiplexing)

ldr r6, =DMAMUX1_BASE // Load reg base address to r6

mov r5,#46 // USART3_TX DMA Device Nr. is 46
str r5, [r6, DMAMUX1_C0CR] // DMAREQ for Channel 0 to USART3_TX
mov r5,#45 // USART3_Rx DMA Device Nr. is 45
str r5, [r6, DMAMUX1_C1CR] // DMAREQ for Channel 1 to USART3_RX

DMA_INIT (3/3 – USART3+DMA): 

//----------------------------------------------
// ------------- USART3 Settings
ldr r6, =USART3_BASE // Load USART3 BASE address to r1

// Disable USART3
ldr r5, [r6,#USART_CR1] // Read its content to r5
bic r5, #1
str r5, [r6,#USART_CR1] // Store result

// Enable DMA Transmit and Receive for USART3
ldr r5, [r6, #USART_CR3]
orr r5, #(0b11<<6) // Set bits 7 and 6 to enable DMAT and DMAR bits
str r5, [r6,#USART_CR3] // Store result

// Enable USART3
ldr r6, =USART3_BASE // Load USART3 BASE address to r6
ldr r5, [r6,#USART_CR1] // Read its content to r5
orr r5, r5, #1
str r5, [r6,#USART_CR1] // Store result

pop {r5, r6, pc}

DMA – oddaja 
Potrebni koraki za krmiljenje DMA naprave (SND_DMA):

1. Počakaj EN bit = 0 v DMA_SxCR_TX registru (zaključek prejšnjega prenosa)

• DMA_SxCR_TX b0=0 (Stream disabled)

2. Nastavitve naslovov

• DMA_SxPAR_TX naslov DR registra (USART3_BASE+USART_TDR)

• DMA_SxM0AR_TX naslov v pomnilniku v r0

• DMA_SxSNDTR_TX število znakov v r1

3. Brisanje zastavic v statusnem registru DMA_LIFCR :

• Clear all bits for channel #0
• 0b111101 -> DMA_LIFCR

4. Vklop DMA kanala

• DMA_SxCR_TX b0=1 (Stream enabled)

5. Čakanje na konec prenosa :

• USART_ISR TC = 1   (DMA označi konec prenosa)

DMA – oddaja (SND_DMA) 
SND_DMA:
push {r5, r6, lr}

ldr r6, =DMA1_BASE // Load reg base address to r6

WAIT_EN1: // Wait EN bit to become zero
ldr r5, [r6, #DMA_SxCR_TX]
tst r5, #1
bne WAIT_EN1

ldr r6, =DMA1_BASE // Load reg base address to r6
// Transmit (TX) DMA Init
ldr r5, =USART3_BASE+USART_TDR // RX peripheral address to r5
str r5, [r6,#DMA_SxPAR_TX] // Store result in peripheral DMA 
pointerclock register
str r0, [r6,#DMA_SxM0AR_TX] // Store address pointer
str r1, [r6,#DMA_SxSNDTR_TX] // Store result in peripheral DMA 
pointerclock register

// Clear flags in Status register
mov r5, #0b111101 // Clear all bits for channel 0
str r5, [r6,#DMA_LIFCR] // Store result in peripheral DMA 
pointerclock register

// Enable DMA Channel
ldr r6, =DMA1_BASE // Load reg base address to r6
ldr r5, [r6, #DMA_SxCR_TX]
orr r5, r5, #1
str r5, [r6, #DMA_SxCR_TX] // Enable channel

// Wait for the end of transmission
ldr r6, =USART3_BASE

WAIT_TC:
ldr r5, [r6, #USART_ISR]
tst r5, #(1 << 6) // Test TC bit
beq WAIT_TC

pop {r5, r6, pc}

DMA – sprejem
Potrebni koraki za krmiljenje DMA naprave (RCV_DMA):

1. Počakaj EN bit = 0 v DMA_SxCR_RX registru (zaključek prejšnjega prenosa)

• DMA_SxCR_RX b0=0 (Stream disabled)

2. Nastavitve naslovov

• DMA_SxPAR_RX naslov DR registra (USART3_BASE+USART_RDR)

• DMA_SxM0AR_RX naslov v pomnilniku v r0

• DMA_SxSNDTR_RX število znakov v r1

3. Brisanje zastavic v statusnem registru DMA_LIFCR :

• Clear all bits for channel #0
• (0b111101<<6) -> DMA_LIFCR

4. Vklop DMA kanala

• DMA_SxCR_RX b0=1 (Stream enabled)

5. Čakanje na konec prenosa :

• DMA_LISR  TCIF1 = 1   (DMA označi konec prenosa)

DMA – sprejem (RCV_DMA) 
RCV_DMA:
push {r5, r6, lr}

ldr r6, =DMA1_BASE // Load reg base address to r6

WAIT_EN: // Wait EN bit to become zero
ldr r5, [r6, #DMA_SxCR_RX]
tst r5, #1
bne WAIT_EN

ldr r6, =DMA1_BASE // Load reg base address to r6

// Receive (RX) DMA Init
ldr r5, =USART3_BASE+USART_RDR // RX peripheral address to r5
str r5, [r6,#DMA_SxPAR_RX] // Store peripheral DMA pointer
str r0, [r6,#DMA_SxM0AR_RX] // Store address pointer
str r1, [r6,#DMA_SxSNDTR_RX] // Store number of units

// Clear flags in Status register
mov r5,#(0b111101<<6) // clear flags for Ch1 in ISR
str r5, [r6,# DMA_LIFCR] // Store

// Enable DMA Channel
ldr r5, [r6, #DMA_SxCR_RX]
orr r5, r5, #1
str r5, [r6, #DMA_SxCR_RX] // Enable channel

// Wait for the end of reception (TCIF
ldr r6, =DMA1_BASE // Load reg base 
address to r6

WAIT_RC:
ldr r5, [r6, #DMA_LISR]
tst r5, #(1 << 11) // TCIF1 flag
beq WAIT_RC

pop {r5, r6, pc}

STM32H7

Vhodno / izhodne naprave

USART Serijska komunikacija

z uporabo DMA krmilnika

Podrobnejši izseki iz dokumentacije 

(Ref.Man. - rm0433)

DMA – sprejem (RM) 

DMA – sprejem (RM) 

DMA – sprejem (RM) 

DMA – oddaja (RM) 

DMA – oddaja (RM) 

DMA – oddaja (RM) 

DMA – oddaja (RM) 

